*==========================================================================================================================================================
*
*	Local Projections
*
*==========================================================================================================================================================


set more off

qui {

*-------------------------------------------------------------------------------------------------------------
*	SETTINGS
*-------------------------------------------------------------------------------------------------------------

* Variables
local vars "gdp_r_adj cpi_sil log_R_esp_er_hp itsN wage_sil " 
local lava ""Real output" "Consumer prices" "Lending rate" "Money growth rate" "Nominal wage"" 
local yti ""Percent" "Percent" "Percentage pts" "Percentage pts" "Percent"" 

* irf horizon
local H 5
local F 5

* lags
local L 2
local LS `L'
*-------------------------------------------------------------------------------------------------------------

local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"

*_________ GET DATA ____________
use "`path'data/working files/prepared.dta", clear 

gen outdum = 0
replace outdum=1 if outbound >0 & outbound!=.

replace outbound = 0 if outbound==.


*____ DECLARE TIME SERIES _____
gen id =1
tsset id year, yearly



*_________________ LOCAL PROJECTIONS _________________

***_________ CONTROLS

local indep	" gdp_r_adj_gr cpi_sil_gr wage_sil_gr itsN_gr" 

forvalues i=0/`L'{
	local indep_`i' = subinstr("`indep'"," "," l`i'_",.)
	}
	
local rhs ""
forvalues i = 1/`L'{
	local rhs "`rhs'`indep_`i''"
	}
	
	local not l0_`v'_gr l0_itsN_gr 
	local rhs: list rhs- not
	


	***_________ IMPULSE RESPONSE and CONFIDENCE INTERVAL variables
	gen irf_`v' = . if _n<=`H'+1
	
	gen se_`v' = . if _n<=`H'+1

	gen ci_up_`v' = . if _n<=`H'+1
	gen ci_lo_`v' = . if _n<=`H'+1
	gen ci_up90_`v' = . if _n<=`H'+1
	gen ci_lo90_`v' = . if _n<=`H'+1
			
	gen N_`v' = . if _n<=`H'+1	

	
	
	***_________ ESTIMATE COEFFICIENTS
	
	forvalues y = 0/`H' {
		n display "projection `y'"

		newey `v'_gr`y' l(1/`LS').`v'_gr l(-`F'/`LS').perclossN_vellon l(-`F'/`LS').perclossN l(-`F'/`LS').percloss_salvN l(-`F'/`LS').percloss_pirN l(-`F'/`LS').outbound  `rhs' l(0/`LS').temperature l(0/`LS').warbeu l(0/`LS').warbciv, lag(5) force 

		* store IMPULSE RESPONSES and STANDARD ERRORS
		lincom outbound 
		replace irf_`v' = r(estimate) if _n==`y'+1 /* coefficients */
		replace se_`v' = r(se) if _n==`y'+1 /* standard errors */
		
		} //horizon			

	

	***_________ CONFIDENCE INTERVALS
	replace ci_up_`v' = irf_`v' + 1*se_`v' if _n <= `H'+1
	replace ci_lo_`v' = irf_`v' - 1*se_`v' if _n <= `H'+1
	replace ci_up90_`v' = irf_`v' + 1.64*se_`v' if _n <= `H'+1
	replace ci_lo90_`v' = irf_`v' - 1.64*se_`v' if _n <= `H'+1

* plot prerequisites
gen time = _n-1 if(_n<=`H'+1) 
drop if time > `H'
keep time irf_* ci_up_* ci_lo_* ci_up90_* ci_lo90_* se_*
order irf_* ci_up_* ci_lo_* time ci_up90_* ci_lo90_*
drop if time == .
gen zero = 0 if time!=.

local title : word `count' of `lava'
local ytitle : word `count' of `yti'

local lsize "3"
local tsize "3.5"
local legsize "3"

sum ci_up90_`v'
local max = round(`r(max)'*1.2,0.5)
if (`max'<0) {
	local max = 0.5
	}

sum ci_lo90_`v'
local min = round(`r(min)'*1.33,0.5)


twoway  (rarea ci_lo90_`v' ci_up90_`v' time, fcolor(gs15) lcolor(gs15)) ///
		(rarea ci_lo_`v' ci_up_`v' time, fcolor(gs12) lcolor(gs12)) ///
		(line irf_`v' time, lcolor(black) lpattern(solid) lwidth(medthick)) ///
		(line zero time, lcolor(black) lwidth(medium)), title("") xtitle("Year", size(`lsize')) ytitle(`ytitle', height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(`min'(1)`max', format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(off) title(`title', size(`tsize')) ///
		graphregion(color(white)) plotregion(color(white)) name(g`count', replace) nodraw
			


} // vars


graph combine g1 g3 g2 g5, name(LP_outbound_presi, replace) cols(2) imargin(medsmall) plotregion(color(white)) graphregion(color(white)) 
graph display LP_outbound_presi, ysize(17) xsize(20)
graph export "results/FigureA7.pdf", replace 

}


*_________________________________________

			
